

## Handout zum Programmierkurs II (Assembler) Quelltextformatierung und Assemblerdirektiven

### Formatierung von Assemblerprogrammen

Seit über 40 Jahren haben sich gewisse Standards etabliert, wie ein Assembler Quellcode zu formatieren ist, unabhängig von der verwendeten Zielplattform. Es wird nahegelegt, sich bei der Programmentwicklung an diesen inoffiziellen Standard zu halten.

- Programmzeilen werden mit einem Tabulator eingerückt
- Sprungmarken stehen in der ersten Spalte, ohne Einrückung, abgeschlossen von einem Doppelpunkt
- Eine Quelltextzeile enthält entweder eine Sprungmarke oder eine Anweisung
- Der Quelltext sollte der Reihe nach enthalten: Variablen- und Konstantendeklarationen, Programmteil, initialisierte Daten, nicht initialisierte Daten

### Beispielquelltext

Hier ein Beispielprogramm, welches die obigen Konzepte zeigt.

```
        ; *** Beispielprogramm zur Benutzung der seriellen Schnittstelle ***
ORG 0100h
        clr ES                ; seriellen Interrupt sperren
        setb TI               ; erfolgreichen Transmit simulieren
        mov dptr,#Teststring ; Stringanfang in den DPTR laden
outloop:
        movx a,@dptr         ; nächstes Zeichen lesen
        jz weiter           ; falls 0, dann aufhören...
        inc dptr             ; sonst: nächstes Zeichen
        lcall SerialOut      ; ... und ausgeben
        ajmp outloop        ; und Ausgabeschleife wiederholen
weiter:
        lcall SerialIn       ; ein Zeichen einlesen
        lcall SerialOut      ; und ausgeben
        ljmp 918eh          ; ... und Programm beenden

        ; *** Unterprogramme ***
        ; lies von der seriellen Schnittstelle...
        ; Zeichen in Akku
SerialIn:
        jnb RI,SerialIn
        clr RI
        mov A,SBUF
        ret
        ; schreib auf die serielle Schnittstelle...
        ; Zeichen im Akku übergeben....
SerialOut:
        jnb TI,SerialOut
        clr TI
        mov SBUF,A
        ret
Teststring:
        DB 27,"[2J",27,"[1mEin fettgedruckter Text",27,"[0m",13,10,0
END
```

Das obige Programm ist kommentiert und hält sich an die Vorgaben, wie Labels und der Programmtext zu formatieren sind.

Natürlich muss man sich nicht sklavisch daran halten, ein derartiges Format ist aber seit langen Jahren üblich und kann leicht von anderen gelesen werden.

### Assemblerdirektiven

Der Assembler *asem-51* kann durch verschiedene Direktiven und sogenannte Pseudoopcodes gesteuert werden. Sie beeinflussen, wie das Programm erzeugt wird. So kann festgelegt werden, ab welcher Adresse das Programm im Speicher stehen wird, unter welcher symbolischen Adresse Strings und Daten abgelegt werden. Auch können symbolische Namen für Variablen, Register und Programmadressen bestimmt werden.

| <i>Direktive</i>               | <i>Erläuterung</i>  |
|--------------------------------|---|
| ORG <adresse>                  | Der folgende Programmcode wird ab der angegebenen Adresse erzeugt. Die Adresse wird üblicherweise als Hexzahl angegeben.  |
| <Bezeichner> EQU <Ersetzung>   | Symbolische Zuweisung bzw. Konstantendeklaration  |
| <Bezeichner> DATA <Adresse>    | Symbolische Zuweisung eines Names für eine Speicherzelle im internen Datenspeicher und für SFRs   |
| <Bezeichner> BIT <Bit-Adresse> | Symbolische Zuweisung eines Names für eine Bit-Adresse  |
| <Bezeichner> XDATA <Adresse>   | vergabe eines symbolischen Namens im externen Datenspeicher   |
| <label>:                       | Sprungmarke für Schleifen und Unterprogramme, symbolische Zuweisung eines Names an einen gegebenen PC-Stand   |
| DB <bytes>,[<strings>,...]     | Fügt Bytes oder ganze Strings in den Codespeicher ein (für String und Datenkonstanten)  |
| DW <words>,[<words>,...]       | Fügt 16Bit lange Wortkonstanten in den Codespeicher ein (für Tabellen aus 16Bit Konstanten)   |
| \$INCLUDE <Datei>              | Externe Datei einfügen (für Konstantendeklarationen, SFRs, etc.)  |
| END                            | Assemblerlauf beenden (letzte Zeile im Quelltext)<br><br><b>Wichtig:</b> Diese Zeile <i>muss</i> zwingend mit einem RETURN bzw. Zeilenvorschub abgeschlossen werden, oder die Direktive wird nicht erkannt. |